

intreg postestimation — Postestimation tools for intreg

[Postestimation commands](#)
 [predict](#)
 [margins](#)
 [Remarks and examples](#)
 Also see

Postestimation commands

The following postestimation commands are available after `intreg`:

Command	Description
<code>contrast</code>	contrasts and ANOVA-style joint tests of estimates
<code>estat ic</code>	Akaike's, consistent Akaike's, corrected Akaike's, and Schwarz's Bayesian information criteria (AIC, CAIC, AICc, and BIC)
<code>estat summarize</code>	summary statistics for the estimation sample
<code>estat vce</code>	variance–covariance matrix of the estimators (VCE)
<code>estat (svy)</code>	postestimation statistics for survey data
<code>estimates</code>	cataloging estimation results
<code>etable</code>	table of estimation results
* <code>hausman</code>	Hausman's specification test
<code>lincom</code>	point estimates, standard errors, testing, and inference for linear combinations of coefficients
* <code>lrtest</code>	likelihood-ratio test
<code>margins</code>	marginal means, predictive margins, marginal effects, and average marginal effects
<code>marginsplot</code>	graph the results from margins (profile plots, interaction plots, etc.)
<code>nlcom</code>	point estimates, standard errors, testing, and inference for nonlinear combinations of coefficients
<code>predict</code>	linear, censored, and truncated predictions
<code>predictnl</code>	point estimates, standard errors, testing, and inference for generalized predictions
<code>pwcompare</code>	pairwise comparisons of estimates
<code>suest</code>	seemingly unrelated estimation
<code>test</code>	Wald tests of simple and composite linear hypotheses
<code>testnl</code>	Wald tests of nonlinear hypotheses

*`hausman` and `lrtest` are not appropriate with `svy` estimation results.

predict

Description for predict

`predict` creates a new variable containing predictions such as linear predictions, standard errors, probabilities, and expected values.

Menu for predict

Statistics > Postestimation

Syntax for predict

```
predict [type] newvar [if] [in] [, statistic nooffset]
```

```
predict [type] stub* [if] [in], scores
```

<i>statistic</i>	Description
------------------	-------------

Main

<code>xb</code>	linear prediction; the default
<code>stdp</code>	standard error of the prediction
<code>stdf</code>	standard error of the forecast
<code>pr(<i>a</i>,<i>b</i>)</code>	$\Pr(a < y_j < b)$
<code>e(<i>a</i>,<i>b</i>)</code>	$E(y_j a < y_j < b)$
<code>ystar(<i>a</i>,<i>b</i>)</code>	$E(y_j^*), y_j^* = \max\{a, \min(y_j, b)\}$

These statistics are available both in and out of sample; type `predict ... if e(sample) ...` if wanted only for the estimation sample.

`stdf` is not allowed with `svy` estimation results.

where *a* and *b* may be numbers or variables; *a* missing (*a* ≥ .) means $-\infty$, and *b* missing (*b* ≥ .) means $+\infty$; see [U] [12.2.1 Missing values](#).

Options for predict

Main

`xb`, the default, calculates the linear prediction.

`stdp` calculates the standard error of the prediction, which can be thought of as the standard error of the predicted expected value or mean for the observation's covariate pattern. The standard error of the prediction is also referred to as the standard error of the fitted value.

`stdf` calculates the standard error of the forecast, which is the standard error of the point prediction for 1 observation. It is commonly referred to as the standard error of the future or forecast value. By construction, the standard errors produced by `stdf` are always larger than those produced by `stdp`; see *Methods and formulas* in [R] [regress postestimation](#).

`pr(a,b)` calculates $\Pr(a < \mathbf{x}_j\boldsymbol{\beta} + \epsilon_j < b)$, the probability that $y_j | \mathbf{x}_j$ would be observed in the interval (*a*, *b*).

a and b may be specified as numbers or variable names; lb and ub are variable names;

`pr(20,30)` calculates $\Pr(20 < \mathbf{x}_j\boldsymbol{\beta} + \epsilon_j < 30)$;

`pr(lb,ub)` calculates $\Pr(lb < \mathbf{x}_j\boldsymbol{\beta} + \epsilon_j < ub)$; and

`pr(20,ub)` calculates $\Pr(20 < \mathbf{x}_j\boldsymbol{\beta} + \epsilon_j < ub)$.

a missing ($a \geq .$) means $-\infty$; `pr(.,30)` calculates $\Pr(-\infty < \mathbf{x}_j\boldsymbol{\beta} + \epsilon_j < 30)$;

`pr(lb,30)` calculates $\Pr(-\infty < \mathbf{x}_j\boldsymbol{\beta} + \epsilon_j < 30)$ in observations for which $lb \geq .$

and calculates $\Pr(lb < \mathbf{x}_j\boldsymbol{\beta} + \epsilon_j < 30)$ elsewhere.

b missing ($b \geq .$) means $+\infty$; `pr(20,.)` calculates $\Pr(+\infty > \mathbf{x}_j\boldsymbol{\beta} + \epsilon_j > 20)$;

`pr(20,ub)` calculates $\Pr(+\infty > \mathbf{x}_j\boldsymbol{\beta} + \epsilon_j > 20)$ in observations for which $ub \geq .$

and calculates $\Pr(20 < \mathbf{x}_j\boldsymbol{\beta} + \epsilon_j < ub)$ elsewhere.

`e(a,b)` calculates $E(\mathbf{x}_j\boldsymbol{\beta} + \epsilon_j \mid a < \mathbf{x}_j\boldsymbol{\beta} + \epsilon_j < b)$, the expected value of $y_j \mid \mathbf{x}_j$ conditional on $y_j \mid \mathbf{x}_j$ being in the interval (a, b) , meaning that $y_j \mid \mathbf{x}_j$ is truncated.

a and b are specified as they are for `pr()`.

`ystar(a,b)` calculates $E(y_j^*)$, where $y_j^* = a$ if $\mathbf{x}_j\boldsymbol{\beta} + \epsilon_j \leq a$, $y_j^* = b$ if $\mathbf{x}_j\boldsymbol{\beta} + \epsilon_j \geq b$, and $y_j^* = \mathbf{x}_j\boldsymbol{\beta} + \epsilon_j$ otherwise, meaning that y_j^* is censored. a and b are specified as they are for `pr()`.

`nooffset` is relevant only if you specified `offset(varname)`. It modifies the calculations made by `predict` so that they ignore the offset variable; the linear prediction is treated as $\mathbf{x}_j\boldsymbol{\beta}$ rather than as $\mathbf{x}_j\boldsymbol{\beta} + \text{offset}_j$.

`scores` calculates equation-level score variables.

The first new variable will contain $\partial \ln L / \partial (\mathbf{x}_j\boldsymbol{\beta})$.

The second new variable will contain $\partial \ln L / \partial \ln \sigma$.

margins

Description for margins

`margins` estimates margins of response for linear predictions, probabilities, and expected values.

Menu for margins

Statistics > Postestimation

Syntax for margins

```
margins [marginlist] [, options]
```

```
margins [marginlist] , predict(statistic ...) [predict(statistic ...) ...] [options]
```

<i>statistic</i>	Description
<code>xb</code>	linear prediction; the default
<code>pr(<i>a,b</i>)</code>	$\Pr(a < y_j < b)$
<code>e(<i>a,b</i>)</code>	$E(y_j a < y_j < b)$
<code><u>y</u>star(<i>a,b</i>)</code>	$E(y_j^*), y_j^* = \max\{a, \min(y_j, b)\}$
<code>stdp</code>	not allowed with <code>margins</code>
<code>stdf</code>	not allowed with <code>margins</code>

Statistics not allowed with `margins` are functions of stochastic quantities other than `e(b)`.

For the full syntax, see [R] [margins](#).

Remarks and examples

stata.com

▷ Example 1: Marginal predictions

Continuing with [example 1](#) of [R] `intreg`, we compute women's expected wages conditional on a woman's wage being higher than \$5,000. To do this, we can use the `e(a,b)` option.

```
. use https://www.stata-press.com/data/r18/womenwage2
(Wages of women, fictional data)
. intreg wage1 wage2 age c.age#c.age i.nev_mar i.rural school tenure
(output omitted)
. predict w1, e(5,.)
. summarize w1
```

Variable	Obs	Mean	Std. dev.	Min	Max
w1	488	18.02362	4.583738	8.717687	35.31161

The predicted wages range from \$8,718 to \$35,312.


We can also examine whether the probability of earning more than \$5,000 varies with age. We can use `margins` to compute the marginal means of the predicted probabilities at different ages.

```
. margins, predict(pr(5,.)) at(age=(20(5)50))
Predictive margins                                Number of obs = 488
Model VCE: OIM
Expression: Pr(y>5), predict(pr(5,.))
1._at: age = 20
2._at: age = 25
3._at: age = 30
4._at: age = 35
5._at: age = 40
6._at: age = 45
7._at: age = 50
```

	Delta-method				
	Margin	std. err.	z	P> z	[95% conf. interval]
_at					
1	.8912598	.0151773	58.72	0.000	.8615127 .9210068
2	.9104568	.0103467	87.99	0.000	.8901775 .930736
3	.9160005	.0120025	76.32	0.000	.892476 .9395251
4	.9096667	.0136693	66.55	0.000	.8828753 .9364581
5	.8894289	.0206992	42.97	0.000	.8488593 .9299985
6	.8491103	.0447429	18.98	0.000	.7614159 .9368048
7	.7781644	.0970557	8.02	0.000	.5879387 .9683902

We can visualize these results by using `marginsplot`:

```
. marginsplot
Variables that uniquely identify margins: age
```



The probability increases until age 30 and decreases thereafter.

4

Also see

[R] [intreg](#) — Interval regression

[U] [20 Estimation and postestimation commands](#)

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).